

NAG Library Routine Document

G13AJF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13AJF applies a fully specified seasonal ARIMA model to an observed time series, generates the state set for forecasting and (optionally) derives a specified number of forecasts together with their standard deviations.

2 Specification

```

SUBROUTINE G13AJF(MR, PAR, NPAR, C, KFC, X, NX, RMS, ST, IST, NST, NFV,
1          FVA, FSD, IFV, ISF, W, IW, IFAIL)

  INTEGER          MR(7), NPAR, KFC, NX, IST, NST, NFV, IFV, ISF(4), IW,
1          IFAIL
  double precision PAR(NPAR), C, X(NX), RMS, ST(IST), FVA(IFV), FSD(IFV),
1          W(IW)

```

3 Description

The time series $x_1, x_2, \dots, x_n$ supplied to the routine is assumed to follow a seasonal autoregressive integrated moving average (ARIMA) model with known parameters.

The model is defined by the following relations.

(a) $\nabla^d \nabla_s^D x_t - c = w_t$ where $\nabla^d \nabla_s^D x_t$ is the result of applying non-seasonal differencing of order d and seasonal differencing of seasonality s and order D to the series x_t , and c is a constant.

(b) $w_t = \Phi_1 w_{t-s} + \Phi_2 w_{t-2s} + \dots + \Phi_P w_{t-Ps} + e_t - \Theta_1 e_{t-s} - \Theta_2 e_{t-2s} - \dots - \Theta_Q e_{t-Qs}$.

This equation describes the seasonal structure with seasonal period s ; in the absence of seasonality it reduces to $w_t = e_t$.

(c) $e_t = \phi_1 e_{t-1} + \phi_2 e_{t-2} + \dots + \phi_p e_{t-p} + a_t - \theta_1 a_{t-1} - \theta_2 a_{t-2} - \dots - \theta_q a_{t-q}$.

This equation describes the non-seasonal structure.

Given the series, the constant c , and the model parameters Φ , Θ , ϕ , θ , the routine computes the following.

(a) The state set required for forecasting. This contains the minimum amount of information required for forecasting and comprises:

(i) the differenced series w_t , for $(N - s \times P) \leq t \leq N$;

(ii) the $(d + D \times s)$ values required to reconstitute the original series x_t from the differenced series w_t ;

(iii) the intermediate series e_t , for $N - \max(p, Q \times s) < t \leq N$;

(iv) the residual series a_t , for $(N - q) < t \leq N$, where $N = n - (d + D \times s)$.

(b) A set of L forecasts of x_t and their estimated standard errors, s_t , for $t = n + 1, n + 2, \dots, n + L$ (L may be zero).

The forecasts and estimated standard errors are generated from the state set, and are identical to those that would be produced from the same state set by G13AHF.

Use of G13AJF should be confined to situations in which the state set for forecasting is unknown. Forecasting from the series requires recalculation of the state set and this is relatively expensive.

4 References

Box G E P and Jenkins G M (1976) *Time Series Analysis: Forecasting and Control* (Revised Edition) Holden-Day

5 Parameters

- 1: MR(7) – INTEGER array *Input*
On entry: the orders vector (p, d, q, P, D, Q, s) of the ARIMA model, in the usual notation.
Constraints:

$$\begin{aligned} p, d, q, P, D, Q, s &\geq 0; \\ p + q + P + Q &> 0; \\ s &\neq 1; \\ \text{if } s = 0, P + D + Q &= 0; \\ \text{if } s > 1, P + D + Q &> 0. \end{aligned}$$
- 2: PAR(NPAR) – *double precision* array *Input*
On entry: the p values of the ϕ parameters, the q values of the θ parameters, the P values of the Φ parameters, and the Q values of the Θ parameters, in that order.
- 3: NPAR – INTEGER *Input*
On entry: The exact number of ϕ , θ , Φ and Θ parameters.
Constraint: $\text{NPAR} = p + q + P + Q$.
- 4: C – *double precision* *Input*
On entry: c , the expected value of the differenced series (i.e., c is the constant correction). Where there is no constant term, C must be set to 0.0.
- 5: KFC – INTEGER *Input*
On entry: must be set to 0 if C was not estimated, and 1 if C was estimated. This is irrespective of whether or not $C = 0.0$. The only effect is that the residual degrees of freedom are one greater when $\text{KFC} = 0$. Assuming the supplied time series to be the same as that to which the model was originally fitted, this ensures an unbiased estimate of the residual mean-square.
Constraint: $\text{KFC} = 0$ or 1.
- 6: X(NX) – *double precision* array *Input*
On entry: the n values of the original, undifferenced time series.
- 7: NX – INTEGER *Input*
On entry: n , the length of the original, undifferenced time series.
- 8: RMS – *double precision* *Output*
On exit: the residual variance (mean square) associated with the model.
- 9: ST(IST) – *double precision* array *Output*
On exit: the NST values of the state set.

- 10: IST – INTEGER *Input*
On entry: the dimension of the array ST as declared in the (sub)program from which G13AJF is called.
Constraint: $IST \geq (P \times s) + d + (D \times s) + q + \max(p, Q \times s)$. The expression on the right-hand side of the inequality is returned in NST.
- 11: NST – INTEGER *Output*
On exit: the number of values in the state set array ST.
- 12: NFV – INTEGER *Input*
On entry: the required number of forecasts. If $NFV \leq 0$, no forecasts will be computed.
- 13: FVA(IFV) – **double precision** array *Output*
On exit: if $NFV > 0$, FVA contains the NFV forecast values relating to the original undifferenced time series.
- 14: FSD(IFV) – **double precision** array *Output*
On exit: if $NFV > 0$, FSD contains the estimated standard errors of the NFV forecast values.
- 15: IFV – INTEGER *Input*
On entry: the dimension of the arrays FVA and FSD as declared in the (sub)program from which G13AJF is called.
Constraint: $IFV \geq \max(1, NFV)$.
- 16: ISF(4) – INTEGER array *Output*
On exit: contains validity indicators, one for each of the four possible parameter types in the model (autoregressive, moving average, seasonal autoregressive, seasonal moving average), in that order.
Each indicator has the interpretation:
–1 On entry the set of parameter values of this type does not satisfy the stationarity or invertibility test conditions.
0 No parameter of this type is in the model.
1 Valid parameter values of this type have been supplied.
- 17: W(IW) – **double precision** array *Workspace*
18: IW – INTEGER *Input*
On entry: the dimension of the array W as declared in the (sub)program from which G13AJF is called.
Constraint: $IW \geq 6 \times n + 5 \times (p + q + P + Q) + Q'^2 + 11 \times Q' + 3 \times P' + 7$,
where $Q' = Q \times s + q$ and $P' = P \times s + p$.
- 19: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $NPAR \neq p + q + P + Q$,
or the orders vector MR is invalid (check the constraints in Section 5),
or $KFC \neq 0$ or 1.

$IFAIL = 2$

On entry, $NX - d - D \times s \leq NPAR + KFC$, i.e., the number of terms in the differenced series is not greater than the number of parameters in the model. The model is over-parameterised.

$IFAIL = 3$

On entry, the workspace array W is too small.

$IFAIL = 4$

On entry, the state set array ST is too small. It must be at least as large as the exit value of NST.

$IFAIL = 5$

This indicates a failure in F04ASF which is used to solve the equations giving estimates of the backforecasts.

$IFAIL = 6$

On entry, valid values were not supplied for all parameter types in the model. Inspect array ISF for further information on the parameter type(s) in error.

$IFAIL = 7$

On entry, $IFV < \max(1, NFV)$.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by G13AJF is approximately proportional to n and the square of the number of backforecasts derived.

9 Example

The data is that used in the example program for G13AFF. Five forecast values and their standard errors, together with the state set, are computed and printed.

9.1 Program Text

```
*      G13AJF Example Program Text
*      Mark 14 Revised. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NPMAX, ISTMAX, IFVMAX, NXMAX, IW
      PARAMETER        (NPMAX=10,ISTMAX=10,IFVMAX=10,NXMAX=30,IW=250)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
```

```

      DOUBLE PRECISION C, RMS
      INTEGER          I, IFAIL, IFV, IST, KFC, NFV, NPAR, NST, NX
*    .. Local Arrays ..
      DOUBLE PRECISION FSD(IFVMAX), FVA(IFVMAX), PAR(NPMAX), ST(ISTMAX),
+      W(IW), X(NXMAX)
      INTEGER          ISF(4), MR(7)
*    .. External Subroutines ..
      EXTERNAL          G13AJF
*    .. Intrinsic Functions ..
      INTRINSIC          MAX
*    .. Executable Statements ..
      WRITE (NOUT,*) 'G13AJF Example Program Results'
*    Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX, (MR(I),I=1,7), NFV
      NPAR = MR(1) + MR(3) + MR(4) + MR(6)
      IF (NX.GT.0 .AND. NX.LE.NXMAX .AND. NPAR.GT.0 .AND. NPAR.LE.NPMAX)
+      THEN
          READ (NIN,*) (X(I),I=1,NX)
          READ (NIN,*) (PAR(I),I=1,NPAR)
          READ (NIN,*) KFC, C
          IST = MR(4) + MR(7) + MR(2) + MR(5) + MR(3) + MAX(MR(1),MR(6)
+      *MR(7))
          IFV = MAX(1,NFV)
          IFAIL = 0
*
          CALL G13AJF(MR,PAR,NPAR,C,KFC,X,NX,RMS,ST,IST,NST,NFV,FVA,FSD,
+      IFV,ISF,W,IW,IFAIL)
*
          WRITE (NOUT,*)
          WRITE (NOUT,99999) 'The residual mean square is ', RMS
          WRITE (NOUT,*)
          WRITE (NOUT,99998) 'The state set consists of ', NST, ' values'
          WRITE (NOUT,99997) (ST(I),I=1,NST)
          WRITE (NOUT,*)
          WRITE (NOUT,99996) 'The ', NFV,
+      ' forecast values and standard errors are -'
          WRITE (NOUT,99995) (FVA(I),FSD(I),I=1,NFV)
      END IF
*
99999 FORMAT (1X,A,F9.2)
99998 FORMAT (1X,A,I1,A)
99997 FORMAT (1X,4F11.4)
99996 FORMAT (1X,A,I2,A)
99995 FORMAT (10X,2F10.2)
      END

```

9.2 Program Data

G13AJF Example Program Data

```

30 1 1 2 0 0 0 0 5
-217 -177 -166 -136 -110 -95 -64 -37 -14 -25
-51 -62 -73 -88 -113 -120 -83 -33 -19 21
17 44 44 78 88 122 126 114 85 64
-0.0547 -0.5568 -0.6636
1 9.9807

```

9.3 Program Results

G13AJF Example Program Results

The residual mean square is 375.91

The state set consists of 4 values

```
64.0000 -30.9807 -20.4495 -2.7212
```

The 5 forecast values and standard errors are -

60.59	19.39
69.50	34.99
79.54	54.25
89.51	67.87
99.50	79.20
